Math 262a, Fall 1999, Glenn Tesler
Using commutative Grobner bases to solve a system of polynomial equations via elimination ideals
12/2/99

```
> restart;
> with(Groebner): with(Mgfun): with(Holonomy):
  with(Ore_algebra):
```

Example 1

We want to solve the following system of equations. Rewrite each equation A=B as A-B=0, and then form an ideal of the A-B 's.

```
> eqs0 := { x+y^2+z^2 = 1 , x^2+y+z^2 = 1 , x^2+y^2+z = 1
  };
  eqs := map(eq -> lhs(eq)-rhs(eq), eqs0);
```

$$eqs0 := \{x+y^2+z^2=1, x^2+y+z^2=1, x^2+y^2+z=1\}$$

$$eqs := \{x^2+y+z^2-1, x^2+y^2+z-1, x+y^2+z^2-1\}$$

To solve it the traditional way, we would try to substitute, say, x in terms of y and z, to get equations only involving y and z; then we would try to substitute, say, y in terms of z, to get equations only involving z. Then we would solve for z, back substitute this into the yz equations and solve for y's, and then back substitute this into the full equations and solve for x's.
Each substitution step produces polynomials in x,y,z (or possibly radicals that can be turned into polynomials).
We construct a Grobner basis GB with lex order x>y>z, called an "elimination order".
If there exists a polynomial f(z) in <eqs> only containing z's, its lead term is divisible by the lead term of some element g of GB; the lead term of g is a power of z, and all other terms in g are smaller than this in lex order, hence also only contain z, so g is purely a function of z. In fact, a Grobner basis of I[z]=<eqs> intersect k[z], is given by GB intersect k[z]. Note this is not true of x or y, because if the lead term is a power of x, the smaller terms could involve y or z.
Continuing this way: let I[z], I[y,z], I[x,y,z] be the subset of I=<eqs> only involving the indicated variables. Then a Grobner basis for I[z] is GB intersect k[z], etc.

```
> A := poly_algebra(x,y,z);
```

$$A := Ore\_algebra$$

```
> T := termorder(A,plex(x,y,z));
```

$$T := term\_order$$

Get the Grobner basis in lex order x>y>z:

```
> GB := gbasis(eqs,T);
```

$$GB := [-2\,z^4 + 11\,z^3 - 6\,z^2 + 8\,z^6 - 12\,z^5 + z,\, z + 4\,z^4 - z^2 + 4\,y\,z^2 - 2\,y\,z - 4\,z^3,\, -y - z^2 + y^2 + z,\, x + y + 2\,z^2 - z - 1]$$

```
> map(factor,GB);
```

$$[z\,(z-1)\,(z+1)\,(2\,z-1)^3,\, z\,(2\,z-1)\,(2\,z^2 - z - 1 + 2\,y),\, (y - 1 + z)\,(y - z),\, x + y + 2\,z^2 - z - 1]$$

```
>
>
```

This induces Grobner bases for I[z] and I[y,z]:

```
> GBz := select(eq -> not(has(eq,y) or has(eq,x)), GB);
  GByz := select(eq -> not(has(eq,x)),GB);
```

$$GBz := [-2\,z^4 + 11\,z^3 - 6\,z^2 + 8\,z^6 - 12\,z^5 + z]$$

$$GByz := [-2\,z^4 + 11\,z^3 - 6\,z^2 + 8\,z^6 - 12\,z^5 + z,\, z + 4\,z^4 - z^2 + 4\,y\,z^2 - 2\,y\,z - 4\,z^3,\, -y - z^2 + y^2 + z]$$

Solve for possible z's:

```
> zsols := {solve({op(GBz)},z)};
```

$$zsols := \{\{z = 1\}, \{z = \text{-}1\}, \{z = \frac{1}{2}\}, \{z = 0\}\}$$

For each found z, find all possible y's by using the equations only involving y and z. Then for each potential (y,z), find all x's.

```
> sols := NULL:
  for zsol in zsols do
       lprint(‘Trying‘,zsol);
       ysols := {solve({op(subs(zsol,GByz))},y)};
       for ysol in ysols do
           lprint(‘   Trying‘,zsol,ysol);
           xsols := {solve({op(subs(zsol,ysol,GB))},x)};
           if xsols<>{} then
             print(‘Found solutions:‘,zsol,ysol,xsols);
             sols := sols, op(map(xsol ->
  {op(xsol),op(ysol),op(zsol)},xsols)):
           fi
       od
  od;
```

```
  sols;
Trying   {z = 1}
```

$$ysols := \{\{y = 0\}\}$$

```
   Trying   {z = 1}   {y = 0}
```

$$Found\ solutions:, \{z = 1\}, \{y = 0\}, \{\{x = 0\}\}$$

```
Trying   {z = -1}
```

$$ysols := \{\{y = -1\}\}$$

```
   Trying   {z = -1}   {y = -1}
```

$$Found\ solutions:, \{z = -1\}, \{y = -1\}, \{\{x = -1\}\}$$

```
Trying   {z = 1/2}
```

$$ysols := \{\{y = \frac{1}{2}\}\}$$

```
   Trying   {z = 1/2}   {y = 1/2}
```

$$Found\ solutions:, \{z = \frac{1}{2}\}, \{y = \frac{1}{2}\}, \{\{x = \frac{1}{2}\}\}$$

```
Trying   {z = 0}
```

$$ysols := \{\{y = 0\}, \{y = 1\}\}$$

```
   Trying   {z = 0}   {y = 0}
```

$$Found\ solutions:, \{z = 0\}, \{y = 0\}, \{\{x = 1\}\}$$

```
   Trying   {z = 0}   {y = 1}
```

$$Found\ solutions:, \{z = 0\}, \{y = 1\}, \{\{x = 0\}\}$$

$$\{z = 1, y = 0, x = 0\}, \{z = -1, x = -1, y = -1\}, \{z = \frac{1}{2}, y = \frac{1}{2}, x = \frac{1}{2}\},$$

$$\{y = 0, z = 0, x = 1\}, \{x = 0, z = 0, y = 1\}$$

```
>
```

Example 2: equations not solvable by radicals

We begin with a system of equations similar to Example 1.

```
> eqs := { x^5+y^2+z^2-4, x^2+2*y^2-5, x*z-1 };
  A := poly_algebra(x,y,z);
  T := termorder(A,plex(x,y,z));
  GB := gbasis(eqs,T);
```

$$eqs := \{x\,z - 1, x^5 + y^2 + z^2 - 4, x^2 + 2\,y^2 - 5\}$$

$$A := Ore\_algebra$$

$$T := term\_order$$

$$GB := [2 + 2\,z^7 - 3\,z^5 - z^3, 4\,y^2 - 2\,z^5 + 3\,z^3 + z - 10, 2\,x + 2\,z^6 - 3\,z^4 - z^2]$$

```
> GByz := select(eq -> not has(eq,x), GB);
  GBz := select(eq -> not has(eq,y),GByz);
```

$$GByz := [2 + 2\,z^7 - 3\,z^5 - z^3,\ 4\,y^2 - 2\,z^5 + 3\,z^3 + z - 10]$$

$$GBz := [2 + 2\,z^7 - 3\,z^5 - z^3]$$

Solve for possible z's:

```
> zsols := {solve({op(GBz)},z)};
```

$$zsols := \{\{z = \mathrm{RootOf}(2\,\_Z^6 + 2\,\_Z^5 - \_Z^4 - \_Z^3 - 2\,\_Z^2 - 2\,\_Z - 2)\},\ \{z = 1\}\}$$

This can't be solved by radicals! It could be done by numerical approximation, but that's anathema to a symbollic computation system, so Maple is able to work with symbollic solutions, i.e., it works in an extension field of the form k[_Z]/(2 _Z^6 + 2 _Z^5 - _Z^4 - _Z^3 - 2 _Z^2 - 2 _Z - 2).

```
> sols := NULL:
  for zsol in zsols do
       lprint(‘Trying‘,zsol);
       ysols := {solve({op(subs(zsol,GByz))},y)};
       for ysol in ysols do
           lprint(‘   Trying‘,zsol,ysol);
           xsols := {solve({op(subs(zsol,ysol,GB))},x)};
           if xsols<>{} then
             print(‘Found solutions:‘,zsol,ysol,xsols);
             sols := sols, op(map(xsol ->
  {op(xsol),op(ysol),op(zsol)},xsols)):
           fi
       od
  od;
  ’sols’ = sols;
Trying   {z = RootOf(2*_Z^6+2*_Z^5-_Z^4-_Z^3-2*_Z^2-2*_Z-2)}
```

$$ysols := \{\{y = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 2\,\%1^5 + 3\,\%1^3 + \%1 - 10)\}\}$$

$$\%1 := \mathrm{RootOf}(2\,\_Z^6 + 2\,\_Z^5 - \_Z^4 - \_Z^3 - 2\,\_Z^2 - 2\,\_Z - 2)$$

```
   Trying   {z = RootOf(2*_Z^6+2*_Z^5-_Z^4-_Z^3-2*_Z^2-2*_Z-2)}   {y = 1/2*RootOf(
_Z^2-2*RootOf(2*_Z^6+2*_Z^5-_Z^4-_Z^3-2*_Z^2-2*_Z-2)^5+3*RootOf(2*_Z^6+2*_Z^5-_Z^4
-_Z^3-2*_Z^2-2*_Z-2)^3+RootOf(2*_Z^6+2*_Z^5-_Z^4-_Z^3-2*_Z^2-2*_Z-2)-10)}
```

$$\mathit{Found\ solutions:},\ \{z = \%1\},\ \{y = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 2\,\%1^5 + 3\,\%1^3 + \%1 - 10)\},$$

$$\{\{x = -1 + \%1^5 + \%1^4 - \frac{1}{2}\,\%1^3 - \frac{1}{2}\,\%1^2 - \%1\}\}$$

$$\%1 := \mathrm{RootOf}(2\,\_Z^6 + 2\,\_Z^5 - \_Z^4 - \_Z^3 - 2\,\_Z^2 - 2\,\_Z - 2)$$

```
Trying   {z = 1}
```

$$ysols := \{\{y = \mathrm{RootOf}(\_Z^2 - 2)\}\}$$

```
   Trying   {z = 1}   {y = RootOf(_Z^2-2)}
```

$$\textit{Found solutions:}, \{z = 1\}, \{y = \mathrm{RootOf}(\_Z^2 - 2)\}, \{\{x = 1\}\}$$

$$sols = \left(\{z = \%1, x = -1 + \%1^5 + \%1^4 - \frac{1}{2}\%1^3 - \frac{1}{2}\%1^2 - \%1, \right.$$

$$y = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 2\,\%1^5 + 3\,\%1^3 + \%1 - 10)\},$$

$$\left. \{z = 1, y = \mathrm{RootOf}(\_Z^2 - 2), x = 1\} \right)$$

$$\%1 := \mathrm{RootOf}(2\,\_Z^6 + 2\,\_Z^5 - \_Z^4 - \_Z^3 - 2\,\_Z^2 - 2\,\_Z - 2)$$

This should be interpreted with multiplicity, i.e., _Z runs over all possible roots of the given expression.